DOCUMENT RESUME

ED 272 532 TM 860 445

AUTHOR Cobern, William W.
TITLE A BASIC Microcomputer Program for Estimating Test Reliability.
PUB DATE 23 May 86
NOTE 35p.
AVAILABLE FROM To obtain copies of the computer program send a MS-DOS formated disk to William W. Cobern, Education Dept. Computer Lab, Austin College, Sherman, Texas 75090.
PUB TYPE Reports - Descriptive (141) -- Computer Programs (101)

EDRS PRICE MF01/PC02 Plus Postage.
DESCRIPTORS *Computer Software; Difficulty Level; Educational Research; Equations (Mathematics); *Estimation (Mathematics); Item Analysis; *Mathematical Models; *Microcomputers; Test Items; *Test Reliability
IDENTIFIERS BASIC Programing Language; *Kuder Richardson Reliability Formulas; *Split Half Test Reliability; Statistical Packages

ABSTRACT

This computer program, written in BASIC, performs three different calculations of test reliability; (1) the Kuder-Richardson method; (2); the "common split-half" method; and (3) the Rulon-Guttman split-half method. The program reads sequential access data files for microcomputers that have been set up by statistical packages such as STATPAC. The program is written in MS-DOS BASIC and is intended for use on IBM microcomputers and compatibles. Some of the program's statements may be changed for use on an Apple IIe microcomputer. The bulk of this document contains the main menu program; program flow charts and statements; and lists of variables, arrays, and notations. (GDC)

A BASIC Microcomputer Program for Estimating Test Reliability

Dr. William W. Cobern
Assistant Professor of Education
Education Department Computer Lab
Austin College
Sherman, Texas 75090

23rd May, 1986

(Program copies are avalaible by sending MS-DOS formated disk to the above address)

For sometime now social and behavioral science researchers have used SPSS (Nie, 1975) implemented on mainframe computers for their numerical analysis jobs. For most of these researchers the power of a mainframe far exceeds their needs, nevertheless the turnaround time on their "small" jobs is still significant due to the heavy demand on computer center facilities. With the advent of the micro-computer many of these researchers, particularly those in education have found a machine that is both more convenient and efficient for their numerical analysis needs. Almost immediately statistical packages designed for the micro-computer appeared on the market. Perhaps the best of these is "STATPAC" by Walonic Associates (1986). One difficulty with "STATPAC" and several other micro-computer statistical packages is that they do not do test reliability calculations. Of those that do most offer no choice of reliability type, only the Kuder-Richardson 20 formula is used.

There are in fact many ways of estimating reliability but the four most common are the test-retest method, the parallel-test method, split-half methods, and the Kuder-Richardson method. A discussion of these methods can be found in Magnusson (1967). The purpose of this paper is to present a BASIC program that does three reliability calculations: two split-half methods and the Kuder-Richardson method. The program reads sequential access data files that have been set up by statistical packages such as "STATPAC." The program is written in

MS-DOS BASIC intended for use on IBM-type micro-computers. The program has not been copyrighted and anyone who wishes to use it is free to do so.

Two Split-Half Methods

The split-half methods of estimating reliability are actually forms of the parallel-test method. A single test is split in half to form two parallel tests that theoretically measure the same true scores. It is assumed that the two halves have roughly equivalent means and standard deviations. When splitting a test two factors must be considered, item difficulty and item content. It is assumed in splitting a test that the halves are parallel with regard to these factors. In the computer program that follows a test is split by odd and even items. This insures approximately parallel content but not difficulty. The program assumes that the test is homogeneous with respect to difficulty. If that is not the case then the reliability may be underestimated. In this situation the user may wish to reorder the test items according to difficulty so that the odd-even splitting of items results in more truly parallel tests. This will result in a higher reliability estimate. Lastly, the program also assumes that the test is a power test, i.e. there is no significant time limitation.

In the "common split-half" method a correlation coefficient between the two halves is calculated by the formula:

$$r_{oe} = \frac{\sum X_o X_e - [\sum X_o \sum X_e / R]}{\sqrt{[\sum X_o^2 - (\sum X_o)^2 / R][\sum X_e^2 - (\sum X_e)^2 / R]}}$$

where X_o is the sum of the odd items, X_e is the sum of the even items, and r is the number of items in one split-half test (Glass & Stanley, 1970, p. 114). The reliability estimation is then calculated with the Spearman-Brown formula (Magnusson, 1967, p. 73):

$$r = \frac{nr_{oe}}{1 + (n - 1)r_{oe}}$$

where r_{oe} is the correlation coefficient between the two split-halves and n is the number of times the test is increased in length. In the common split-half method the actual test is twice the size of the split-half, therefore n is always 2.

The second split-half method is the Rulon-Guttman method. This method does not necessarily assume that the split-half tests have equal variances. The reliability estimate is based upon the error variance according to the following formula derived by P. J. Rulon:

$$r = 1 - S_d^2 / S_T^2$$

where S_d^2 is the variance of the differences between odd and even scores (Magnusson, 1967, p. 111). In this program the actual equation used is a refinement of the Rulon equation derived by L. Guttman:

$$r = 4r_{oe} S_o S_e / S_T^2$$

where So is the standard deviation of the odd split-half, Se is the standard deviation of the even split-half, roe is the correlation between the two halves, and S_T^2 is the variance of the whole test (Magnusson, 1967, p. 111). When the variances of the two split-halves are equal then the reliability estimate will be the same as the common split-half estimate. When they are not equal then the common split-half method will systematically give a higher estimate. In this case the Rulon-Guttman method is preferable.

The Kuder-Richardson Method

The Kuder-Richardson method is based on the inter-item homogeneity of a test. It is generally used when a test is designed to measure only one trait. Therefore it is a "random parallel test" estimate of reliability as opposed to "parallel test" estimates derived from the split-half methods. In this program the formula used is:

$$r = \frac{n}{n-1}\left(\frac{S_T^2 - \sum S_i^2}{S_T^2}\right)$$

where n is the number of test items, S_T^2 is the test variance, and S stands for item variance (Magnusson, 1967, p. 116). This formula is commonly called the KR-20 estimate of reliability and should not be confused with the KR-21 formula. The KR-21 formula is a more simple calculation but requires the assumption of item variance homogeneity. When the calculations are done by computer the more simple calculation if of no advantage; and since the KR-20 formula is less restrictive it

is used in this program. In this method as the homogeneity of items increases, so does the reliability estimate.

Program Structure

The computer program actually consists of three programs. The first is a menu program that shows the user what reliability method options are available and asks for the user's choice. The second program is for the Kuder-Richardson method and the third program is for the two split-half methods. A flow chart showing program structure precedes the code listing for each program. The code listing itself contains explanatory comments. As stated previously the code is MS-DOS BASIC. For use on an Apple IIe the file commands, the LPRINT, PRINT USING, CLS and LOCATE statements, and the variable name lengths would need to be changed.

The Main Menu



```
10 CLS
20 REM   "RELIABILITY MAIN MENU PROGRAM"        FILE NAME = RELI
30 REM
40 PRINT "                        RELIABILITY PROGRAMS"
50 PRINT "                        ********************"
60 PRINT
70 PRINT
80 PRINT "       1  ................. Kuder Richardson-20"
90 PRINT
100 PRINT "       2  ................. Spit-half Methods"
110 PRINT
120 PRINT "       3 ................. END PROGRAM"
130 PRINT
140 PRINT
150 PRINT
160 INPUT "Enter Number of Choice:  ",X
170 REM
180 ON X GOTO 200, 220, 240
190 REM
200 LOAD "KR20",R
210 REM
220 LOAD "SHM",R
230 REM
240 CLS
250 SYSTEM
```

The KR-20 Method

NOTATION:

X_i test item value where i is the number of the item

X_i^2 square of test item value

T_r test score, i.e. the sum of X_i, where r is the case #

T_r^2 square of test score

S_i^2 variance of test item

S_t^2 test variance

The sums of X_i, X_i^2, and S_t^2 are held in arrays.

MAIN MODULE: Lines 10 - 530



```
10 CLS
20 REM       "Kuder-Richardson -- 20"          File Name: KR20
30 REM
40 REM
50 REM      VARIABLE LIST:
60 REM                        R             RECORD COUNTER
70 REM                        C             CASE COUNTER
80 REM                        RR            RECORD COUNTER
90 REM                        NVAR          # OF VARIABLES PER RECORD
100 REM                       SumIvar       SUM OF ITEM VARIANCES
110 REM                       Tvar          TEST VARIANCE
120 REM                       SumSqSco      SUM OF SCORES SQUARED
130 REM                       SumScore      SUM OF SCORES
140 REM                       FILEN$        DATA FILE NAME
150 REM                       Y$            CORRECT KEYBOARD INPUT
160 REM                       KR20          RELIABILITY COEFFICIENT
170 REM                       W$            INKEY$ IN OUTPUT MODULE
180 REM                       P$            OUTPUT TO PRINTER
190 REM                       FRSTITEM      1ST TEST ITEM
200 REM                       LASTITEM      LAST TEST ITEM
210 REM                       Z             # OF ITEMS ON TEST
220 REM                       CC            COUNTER FOR NON-TEST VARIABLES
230 REM                       VAR$          DUMMY VAR FOR NON-TEST VARIABLES
240 REM                       VAR           DUMMY VAR FOR TEST ITEMS
250 REM
260 REM
270 REM      ARRAY LIST:
280 REM                       SUM(ITEMS)      SUM OF INDIVIDUAL ITEMS
290 REM                       SCORE(CASES)  SET OF SCORES
300 REM                       SUMSQVAR(ITEMS)   SUM OF ITEM SQUARES
310 REM                       IVAR(ITEMS)     ITEM VARIANCES
```

```
320 REM
330 REM
340 REM ************************************************************************
350 REM                          MAIN MODULE
360 REM ************************************************************************
370 REM
380 GOSUB 1000                              'ENTER KEYBOARD INFO
390 REM
400 PRINT "READING DATA FROM FILE  ";FILEN$
410 GOSUB 2000
420 CLOSE#1
430 IF R<> CASES THEN GOTO 6000          '# OF CASES ERROR
440 PRINT
450 PRINT "CALCULATING VARIANCES"
460 GOSUB 3000
470 PRINT
480 PRINT "CALCULATING KR - 20"
490 GOSUB 4000
500 PRINT
510 GOSUB 5000                              'OUTPUT MODULE
520 REM
530 LOAD "RELI", R
```

GOSUB 1000: Enter Test Information from Keyboard



```
970 REM
980 PRINT "ERROR"
990 STOP
1000 REM ************************************************************************
1010 REM                          ENTER KEYBOARD INFO
1020 REM ************************************************************************
1030 REM
1040 CLS
1050 PRINT "                KUDER-RICHARSON 20 RELIABILITY PROGRAM"
1060 PRINT:PRINT:PRINT
1070 PRINT "ENTER FILE NAME    "
1080 PRINT
1090 PRINT "ENTER NUMBER OF CASES    "
1100 PRINT
1110 PRINT "ENTER NUMBER OF VARIABLES PER CASE  "
1120 PRINT
1130 PRINT "ENTER NUMBER OF FIRST TEST ITEM  "
1140 PRINT
1150 PRINT "ENTER NUMBER OF LAST TEST ITEM  "
1160 LOCATE 5,40:INPUT FILEN$
1170 LOCATE 7,40:INPUT CASES
1180 LOCATE 9,40:INPUT NVAR
1190 LOCATE 11,40:INPUT FRSTITEM
1200 LOCATE 13,40:INPUT LASTITEM
1210 CLS
1220 PRINT USING "FILE NAME =      \           \";FILEN$
1230 PRINT
1240 PRINT USING "No. OF CASES =        #########";CASES
1250 PRINT
1260 PRINT USING "No. OF VAR'S =        #########";NVAR
1270 PRINT
1280 PRINT USING "No. OF FIRST TEST ITEM = ####";FRSTITEM
1290 PRINT
1300 PRINT USING "No. OF LAST TEST ITEM =  ####";LASTITEM
1310 PRINT
1320 LET Z = LASTITEM - FRSTITEM + 1
1330 PRINT USING "No. OF TEST ITEMS =      ####";Z
1340 PRINT:PRINT:PRINT
1350 INPUT "ARE THESE CORRECT (Y/N)";Y$
1360 IF Y$<> "Y" AND Y$ <> "y" THEN GOTO 1040
1370 DIM SUM(Z): DIM SCORE(CASES): DIM SUMSQVAR(Z): DIM IVAR(Z)
1380 RETURN
```


Initialize Variables
R, SUMSQSC
SUMSCORE
Open Data File
Does 1st Var = 1st Test Item?
No
GOSUB 2300
Read Non-Test Var's into Dummy Var
Yes
EOF?
Yes
Return to Main Module
No
Update Case Counter
"FOR...NEXT STRUCTURE"
Last Test Item?
No
Read Test Item X i
Add X i to Score (T r)
Yes
Update Array of ∑X i
Update Array 2 of ∑X i
Update ∑T r and ∑T r 2
Does Last Test Item = Last Var?
Yes
No
GOSUB 2300: Read Non-Test Var's into Dummy Var

```
1970 REM
1980 PRINT "ERROR!!!!"
1990 STOP
2000 REM ************************************************************************
2010 REM                    READ DATA
2020 REM ************************************************************************
2030 REM
2040 LET R = 0: LET SUMSQSCO = 0: LET SUMSCORE = 0
2050 OPEN "I", #1, FILEN$
2060 IF FRSTITEM = 1 THEN GOTO 2090
2070    LET Q = 1: LET W = (FRSTITEM - 1)          'SENT TO SUBROUTINE*
2080    GOSUB 2300                                 'NON-TEST VAR'S
2090 REM
2100 IF EOF(1) THEN RETURN
2110     LET R = R + 1  :PRINT "R = ";R
2120     FOR C = FRSTITEM TO LASTITEM
2130       INPUT #1, VAR
2140       PRINT "   VAR  ";C;" = ";VAR
2150       LET SUM(C) = SUM(C) + VAR
2160       LET SUMSQVAR(C) = SUMSQVAR(C) + VAR^2
2170       LET SCORE(R) = SCORE(R) + VAR
2180     NEXT C
2190     LET SUMSQSCO = SUMSQSCO + SCORE(R)^2
2200     LET SUMSCORE = SUMSCORE + SCORE(R)
2210     REM
2220     IF LASTITEM = NVAR THEN GOTO 2250
2230        LET Q = (LASTITEM + 1): LET W = NVAR
2240        GOSUB 2300
2250     REM
2260 GOTO 2100                    'END OF LOOP
2270 REM
2280 PRINT "ERROR!!!"
2290 STOP
2300 REM                            ********************************************
2310 REM                              READ NON-TEST VARIABLES
2320 REM                            ********************************************
2330 REM
2340 FOR CC = Q TO W
2350   INPUT#1, VAR$
2360 NEXT CC
2370 RETURN
```

* From lines 2070 and 2230 values are sent to the FOR...NEXT loop at lines 2340 to 2360 so that any non-test variables in the data file in front of and/or behind the test items are skipped.

GOSUB 3000: Calculate Variances



GOSUB 4000: Calculate KR-20



** If the sum of the item variances is equal to or greater than twice the test variance then the KR-20 value will be less than -1.0. Therefore the test has zero reliability.

```
2970 REM
2980 PRINT "ERROR!!!!"
2990 STOP
3000 REM ************************************************************************
3010 REM                    CALCULATE VARIANCES
3020 REM ************************************************************************
3030 REM
3040 FOR C = FRSTITEM TO LASTITEM
3050   LET IVAR(C) = (SUMSQVAR(C) - ((SUM(C)^2)/R))/(R-1)
3060   LET SUMIVAR = SUMIVAR + IVAR(C)
3070 NEXT C
3080 REM
3090 FOR RR = 1 TO R
3100   LET TVAR = (SUMSQSCO - ((SUMSCORE^2)/R))/(R-1)
3110 NEXT RR
3120 REM
3130 RETURN
3970 REM
3980 PRINT ERROR!!!!"
3990 STOP
4000 REM ************************************************************************
4010 REM                      CALCULATING KR - 20 COEFFICIENT
4020 REM ************************************************************************
4030 REM
4040 LET KR20 = (Z/(Z-1)) * ((TVAR - SUMIVAR)/TVAR)
4050 IF ABS(KR20) <= 1 THEN RETURN
4060 REM
4070 CLS : PRINT "SUM OF ITEM VARIANCES IS > 2 X TEST VARIANCE"
4080 END
```

GOSUB 5000: Output to Printer/Screen



```
4970 REM
4980 PRINT "ERROR!!!!"
4990 STOP
5000 REM ************************************************************************
5010 REM                        OUTPUT
5020 REM ************************************************************************
5030 PRINT:PRINT:PRINT
5040 PRINT "SEND OUTPUT TO SCREEN ONLY OR ALSO TO PRINTER?
5050 PRINT
5060 INPUT "ENTER S/P  "; P$
5070 IF P$<> "P" AND P$<> "p" AND P$<> "S" AND P$<> "s" THEN GOTO 5060
5080 REM
5090 IF P$ = "S" OR P$ = "s" THEN CLS : GOTO 5280
5100 CLS
5110 LPRINT "************************************************************************
"
5120 LPRINT "            KUDER-RICHARDSON - 20 RELIABILITY COEFFICIENT"
5130 LPRINT "************************************************************************
"
5140 LPRINT
5150 LPRINT
5160 LPRINT USING "DATA FILE NAME     \           \";FILEN$
5170 LPRINT
5180 LPRINT USING "No. OF CASES READ          ######"; R
5190 LPRINT
5200 LPRINT USING "No. OF ITEMS               ######"; Z
5210 LPRINT
5220 LPRINT USING "TEST VARIANCE =          #####.###";TVAR
5230 LPRINT
5240 LPRINT USING "SUM OF ITEM VARIANCES = #####.###";SUMIVAR
5250 LPRINT
5260 LPRINT USING "                    KR-20 =  ##.###";KR20
5270 REM
5280 PRINT "************************************************************************"
5290 PRINT "            KUDER-RICHARDSON - 20 RELIABILITY COEFFICIENT"
5300 PRINT "************************************************************************"
5310 PRINT
5320 PRINT
5330 PRINT USING "DATA FILE NAME     \           \";FILEN$
5340 PRINT
5350 PRINT USING "No. OF CASES READ    ######"; R
5360 PRINT
5370 PRINT USING "No. OF ITEMS         ######"; Z
5380 PRINT
5390 PRINT USING "TEST VARIANCE =          #####.###";TVAR
5400 PRINT
5410 PRINT USING "SUM OF ITEM VARIANCES = #####.###";SUMIVAR
5420 PRINT
5430 PRINT USING "                   KR-20 =  ##.###";KR20
5440 REM
5450 PRINT:PRINT:PRINT:PRINT
5460 INPUT "SEND OUTPUT TO PRINTER (Y/N)";Y$
5470 IF Y$ = "Y" OR Y$ = "y" THEN GOTO 5100
5480 REM
5490 RETURN
```

```
5970 REM
5980 PRINT "ERROR!!!!"
5990 STOP
6000 REM ************************************************************************
6010 REM              "R <> CASES" ERROR
6020 REM ************************************************************************
6030 PRINT "R <> CASES ERROR!!!!"
6040 PRINT
6050 PRINT "R = "; R  ;"AND CASES =  ";CASES
6060 PRINT
6070 STOP
```

Split-Half Methods

Notation:

X_o	odd test item, i.e. 1,3,5,7 etc.
X_o^2	square of odd test item
X_e	even test item, i.e. 2,4,6 etc.
X_e^2	square of even test item
$X_o X_e$	product of even item and odd item
T	score on test (odd items + even)
T^2	square of test score
S_o^2	odd half variance
S_e^2	even half variance
R_{oe}	odd-even correlation coefficient
S_T^2	test variance

R csh — common split-half coefficient

R rg — Rulon-Guttman split-half coefficient

Main Module: Lines 10 - 700



```
10 CLS
20 REM       "Split-Half Methods"          File Name: SHM
30 REM
40 REM
50 REM      VARIABLE LIST:
60 REM                        R             RECORD COUNTER
70 REM                        C             CASE COUNTER
80 REM                        RR            RECORD COUNTER
90 REM                        NVAR          # OF VARIABLES PER RECORD
100 REM                       CASES         # OF CASES OR RECORDS
110 REM                       FRSTITEM      FIRST ITEM OF TEST
120 REM                       LASTITEM      LAST ITEM OF TEST
130 REM                       Z             # OF ITEMS ON TEST
140 REM                       FILEN$        DATA FILE NAME
150 REM                       Y$            CORRECT KEYBOARD INPUT
160 REM                       CSH           COMMON S. H. RELIABILITY COEFFICIENT
170 REM                       RGSH          RULON-GUTTMAN S. H. REL. COEFFICIENT
180 REM                       W$            INKEY$ IN OUTPUT MODULE
190 REM                       P$            OUTPUT TO PRINTER
200 REM                       ITEMS         # OF ITEMS ON TEST
210 REM                       CC            COUNTER FOR NON-TEST VARIABLES
220 REM                       VAR$          DUMMY VAR FOR NON-TEST VARIABLES
230 REM                       SUMODD        SUM OF ODD ITEMS
240 REM                       SUMEVN        SUM OF EVEN ITEMS
250 REM                       SUMODEV       SUM OF ODD ITEM * EVEN ITEM
260 REM                       Q AND W       RECORD COUNTERS FOR NON-TEST VARIABLES
270 REM                       ODDITEM       SCORE OF ODD ITEM
280 REM                       EVNITEM       SCOER OF EVEN ITEM
290 REM                       SMSQODD       SUM OF SQUARE ODD ITEMS
300 REM                       SMSQEVN       SUM OF SQUARE EVEN ITEMS
310 REM                       TESTVARO      VARIANCE OF ODD ITEMS
320 REM                       TESTVARE      VARIANCE OF EVEN ITEMS
330 REM                       COVARAB       COVARIANCE OF ODD AND EVEN ITEMS
340 REM                       ROE           ODD ITEM-EVEN ITEM CORRELATION COEF.
350 REM                       TESTSQSC      SUM OF ALL SQUARED SCORES (ODD + EVEN)
360 REM                       TESTSUSC      SUM OF ALL SCORES (ODD + EVEN)
370 REM                       TVAR          TEST VARIANCE
380 REM                       VARE & VARO       STEP IN RAB CALCULATION
390 REM                       OC & EC           ODD AND EVEN ITEM COUNTERS
400 REM
410 REM      ARRAY LIST:
420 REM                       ODDSCOR(CASES)    SET OF ODD ITEM SCORES
430 REM                       EVNSCOR(CASES)    SET OF EVEN ITEM SCORES
440 REM
450 REM
```

```
500 REM ************************************************************************
510 REM                          MAIN MODULE
520 REM ************************************************************************
530 REM
540 GOSUB 1000                              'ENTER KEYBOARD INFO
550 REM
560 PRINT "READING DATA FROM FILE  ";FILEN$
570 GOSUB 2000
580 CLOSE#1
590 IF R<> CASES THEN GOTO 6030          '# OF CASES ERROR
600 PRINT
610 PRINT "CALCULATING VARIANCES"
620 GOSUB 3000
630 PRINT
640 PRINT "CALCULATING SPLIT-HALF RELIABILITY COEFFICIENTS"
650 GOSUB 4000
660 PRINT
670 GOSUB 5000                              'OUTPUT MODULE
680 REM
690 CLS: LOAD "RELI",R
700 END
```

GOSUB 1000: Enter Test Information from Keyboard

Same as flow chart for KR-20; see page 11.

```
970 REM
980 PRINT "ERROR"
990 STOP
1000 REM ************************************************************************
1010 REM                              ENTER KEYBOARD INFO
1020 REM ************************************************************************
1030 REM
1040 CLS
1050 PRINT "************************************************************************
1060 PRINT "              RELIABILTIY: SPLIT-HALF METHODS
1070 PRINT "************************************************************************
1080 PRINT:PRINT:PRINT
1090 INPUT "ENTER FILE NAME    ",FILEN$
1100 PRINT
1110 INPUT "ENTER NUMBER OF CASES    ",CASES
1120 PRINT
1130 INPUT "ENTER NUMBER OF VARIABLES PER CASE   ",NVAR
1140 PRINT
1150 INPUT "ENTER NUMBER OF FIRST TEST ITEM   ";FRSTITEM
1160 PRINT
1170 INPUT "ENTER NUMBER OF LAST TEST ITEM   "; LASTITEM
1180 LET Z = LASTITEM - FRSTITEM + 1
1190 CLS
1200 PRINT USING "FILE NAME =       \             \";FILEN$
1210 PRINT
1220 PRINT USING "No. OF CASES =    ########";CASES
1230 PRINT
1240 PRINT USING "No. OF VAR'S =    ########";NVAR
1250 PRINT
1260 PRINT USING "FIRST ITEM =         #####";FRSTITEM
1270 PRINT
1280 PRINT USING "LAST ITEM  =         #####";LASTITEM
1290 PRINT
1300 PRINT USING "No. OF ITEMS =       #####";Z
1310 PRINT:PRINT:PRINT
1320 INPUT "ARE THESE CORRECT (Y/N)";Y$
1330 IF Y$<> "Y" AND Y$ <> "y" THEN GOTO 1040
1340 DIM ODDSCOR(CASES): DIM EVNSCOR(CASES)
1350 CLS
1360 RETURN
1970 REM
1980 PRINT "ERROR!!!!"
```

GOSUB 2000: Read Data File




Update:
$\sum X_o^2$ & $\sum X_o$
$\sum X_e^2$ & $\sum X_e$
$\sum X_o X_e$ & $\sum T^2$
Last Item = last Var?
Yes
No
GOSUB 2500
Read Non-Test Items Into Dummy Var's

```
1970 REM
1980 PRINT "ERROR!!!!"
1990 STOP
2000 REM ************************************************************************
2010 REM                    READ DATA
2020 REM ************************************************************************
2030 REM
2040 LET R = 0: LET SMSQODD = 0: LET SMSQEVN = 0
2050 LET SUMODD = 0: LET SUMEVN = 0
2060 LET SUMODEV = 0
2070 OPEN "I", #1, FILEN$
2080 IF FRSTITEM = 1 THEN GOTO 2110
2090    LET Q = 1: LET W = (FRSTITEM - 1)          'SENT TO SUBROUTINE
2100    GOSUB 2500                                 'NON-TEST VAR'S
2110 REM
2120 IF EOF(1) THEN RETURN
2130     LET R = R + 1   :PRINT "R = ";R
2140     LET EC = 0: LET OC = 0
2150     FOR C = FRSTITEM TO LASTITEM
2160       INPUT#1, ODDITEM: PRINT ODDITEM
2170       IF C = LASTITEM THEN GOTO 2200
2180       INPUT#1, EVNITEM: PRINT EVNITEM
2190       LET EVNSCOR(R) = EVNSCOR(R) + EVNITEM:    LET EC = EC + 1
2200       LET ODDSCOR(R) = ODDSCOR(R) + ODDITEM:    LET OC = OC + 1
2210       LET C = C + 1
2220     NEXT C
2230       IF EC <> OC THEN GOSUB 2610
2240       LET SMSQODD = SMSQODD + ODDSCOR(R)^2
2250       LET SUMODD = SUMODD + ODDSCOR(R)
2260       LET SMSQEVN = SMSQEVN + EVNSCOR(R)^2
2270       LET SUMEVN = SUMEVN + EVNSCOR(R)
2280       LET SUMODEV = SUMODEV + (ODDSCOR(R) * EVNSCOR(R))
2290       LET TESTSQSC = TESTSQSC + (ODDSCOR(R) + EVNSCOR(R))^2
2300     REM
2310     IF LASTITEM = NVAR THEN GOTO 2340
2320        LET Q = (LASTITEM + 1): LET W = NVAR
2330        GOSUB 2500
2340     REM
2350 GOTO 2120                    'END OF LOOP
2470 REM
2480 PRINT "ERROR!!!"
2490 STOP
2500 REM                          ***********************************************
2510 REM                            READ NON-TEST VARIABLES
2520 REM                          ***********************************************
2530 REM
2540 FOR CC = Q TO W
2550   INPUT#1, VAR$
2560 NEXT CC
2570 RETURN
2580 REM
2590 PRINT "ERROR!!!!"
2600 STOP
2610 REM                          ***********************************************
2620 REM                            ADJUSTING FOR ODD NUMBER OF ITEMS
2630 REM                          ***********************************************
2640 LET ODDSCOR(R) = ODDSCOR(R)/OC
2650 LET EVNSCOR(R) = EVNSCOR(R)/EC
2660 RETURN
```

GOSUB 3000: Calculate Variances



GOSUB 4000: Calculate Split-Half R's



```
2970 REM
2980 PRINT "ERROR!!!!"
2990 STOP
3000 REM ************************************************************************
3010 REM                    CALCULATE VARIANCES
3020 REM ************************************************************************
3030 REM
3040 REM
3050   LET VARO = SMSQODD - ((SUMODD^2)/R)
3060   LET TESTVARO = VARO/(R-1)
3070   LET VARE = SMSQEVN - ((SUMEVN^2)/R)
3080   LET TESTVARE = VARE/(R-1)
3090   LET COVARAB = SUMODEV - (SUMODD * SUMEVN)/R
3100   LET ROE = COVARAB/ (SQR(VARO) * SQR(VARE))
3110 REM
3120   LET TESTSUSC = SUMODD + SUMEVN
3130   LET TESTVAR = (TESTSQSC - ((TESTSUSC^2)/R))/(R - 1)
3140 RETURN
3970 REM
3980 PRINT ERROR!!!!"
3990 STOP
4000 REM ************************************************************************
4010 REM                      CALCULATING SPLIT-HALF RELIABILITIES
4020 REM ************************************************************************
4030 REM
4040 '                              COMMON SPLIT-HALF METHOD
4050 LET CSH = (2 * ROE)/(1 + ROE)
4060 '                              RULON-GUTTMAN SPLIT-HALF METHOD
4070 LET RG = ROE * SQR(TESTVARO) * SQR(TESTVARE)/TESTVAR
4080 IF ABS(RG) => .25 THEN GOTO 6500
4090 LET RGSH = 4 * RG
4100 RETURN
```

GOSUB 5000: Output to Printer/Screen

Same as flow chart for KR-20; see page 17.

```
4970 REM
4980 PRINT "ERROR!!!!"
4990 STOP
5000 REM ************************************************************************
5010 REM                        OUTPUT
5020 REM ************************************************************************
5030 PRINT:PRINT:PRINT
5040 PRINT "SEND OUTPUT TO SCREEN ONLY OR ALSO TO PRINTER?
5050 PRINT
5060 INPUT "ENTER S/P  "; P$
5070 IF P$<> "P" AND P$<> "p" AND P$<> "S" AND P$<> "s" THEN GOTO 5060
5080 REM
5090 IF P$ = "S" OR P$ = "s" THEN CLS : GOTO 5340
5100 CLS
5110 LPRINT "************************************************************************
"
5120 LPRINT "           SPLIT-HALF RELIABILITY COEFFICIENTS"
5130 LPRINT "************************************************************************
"
5140 LPRINT
5150 LPRINT
5160 LPRINT USING "DATA FILE NAME     \           \";FILEN$
5170 LPRINT
5180 LPRINT USING "No. OF CASES READ    ######"; R
5190 LPRINT
5200 LPRINT USING "No. OF ITEMS         ######"; Z
5210 LPRINT
5220 LPRINT USING "S-ODD  =            ###.###";SQR(VARO)
5230 LPRINT
5240 LPRINT USING "S-EVEN =            ###.###";SQR(VARE)
5250 LPRINT
5260 LPRINT USING "MEAN-ODD  =         ###.###";SUMODD/R
5270 LPRINT
5280 LPRINT USING "MEAN-EVEN =         ###.###";SUMEVN/R
5290 LPRINT
5300 LPRINT USING "COMMON SPLIT-HALF METHOD          r = ##.###";CSH
5310 LPRINT
5320 LPRINT USING "RULON-GUTTMAN SPLIT-HALF METHOD   r = ##.###";RGSH
5330 REM
```

```
5340 PRINT "*************************************************************"
5350 PRINT "          SPLIT-HALF RELIABILITY COEFFICIENTS"
5360 PRINT "*************************************************************"
5370 PRINT
5380 PRINT
5390 PRINT USING "DATA FILE NAME     \          \";FILEN$
5400 PRINT
5410 PRINT USING "No. OF CASES READ    ######"; R
5420 PRINT
5430 PRINT USING "No. OF ITEMS         ######"; Z
5440 PRINT
5450 PRINT USING "S-ODD  =            ###.###";SQR(VARO)
5460 PRINT
5470 PRINT USING "S-EVEN =            ###.###";SQR(VARE)
5480 PRINT
5490 PRINT USING "MEAN-ODD  =         ###.###";SUMODD/R
5500 PRINT
5510 PRINT USING "MEAN-EVEN =         ###.###";SUMEVN/R
5520 PRINT
5530 PRINT USING "COMMON SPLIT-HALF METHOD          r =  ##.###";CSH
5540 PRINT
5550 PRINT USING "RULON-GUTTMAN SPLIT-HALF METHOD   r =  ##.###";RGSH
5560 REM
5570 LOCATE 12,45 : PRINT "ENTER RETURN TO CONTINUE"
5580 LOCATE 12,69 : LET W$ = INKEY$: IF W$ = "" THEN GOTO 5580
5590 REM
5600 RETURN
```

```
5970 REM
5980 PRINT "ERROR!!!!"
5990 STOP
6000 REM ************************************************************************
6010 REM            "R <> CASES" ERROR
6020 REM ************************************************************************
6030 PRINT "R <> CASES ERROR!!!!"
6040 PRINT
6050 PRINT "R = "; R  ;"AND CASES =  ";CASES
6060 PRINT
6070 STOP
6080 REM
6090 REM
6500 REM ************************************************************************
6510 REM       "RAB * TESTVARE * TESTVARO / TESTVAR => 0.25" ERROR
6520 REM ************************************************************************
6530 REM
6540 CLS
6550 PRINT "ABS VALUE [ RAB * TESTVARO * TESTVARE / TESTVAR ] => 0.25" **
6560 PRINT:PRINT
6570 PRINT USING "ROE = ##.###";ROE
6580 PRINT
6590 PRINT USING "VARIANCE OF ODD HALF =  ###.###";TESTVARO
6600 PRINT
6610 PRINT USING "VARIANCE OF EVEN HALF = ###.###";TESTVARE
6620 PRINT
6630 PRINT USING "TEST VARIANCE =  ###.###";TESTVAR
6640 PRINT
6650 PRINT USING "RATIO =  ##.###";ROE*TESTVARO*TESTVARE/TESTVAR
6660 REM
6670 END
```

** The ratio in line 6650 is the value referred to in line 6550. If this happens the program is aborted because the resulting r value would be greater than the absolute value of 1.

References

Magnusson, David. Test Theory. Addison-Wesley Publishing Co., Reading, MA, 1967

Nie, N. J., et al. SPSS. McGraw Hill Book Company, New York, 1975.

Walonick, D. STATPAC Version 6.0. Walonick Associates, Minneapolis, 1986.